
PENAMPIL TOMBOL-TOMBOL REMOTE CONTROL 
SONY PADA M1632 LCD OLEH MODUL DST-52 

  
 Pada artikel-artikel sebelumnya, banyak dibahas penggunaan keypad ataupun PC 
keyboard sebagai media input data dari suatu system berbasis mikrokontroler seperti 
DST-52. Pada artikel ini, akan dibahas satu jenis lagi dari media input yang ada yaitu 
remote control. Dibandingkan dengan PC Keyboard maupun keypad, sebuah remote 
control seringkali dirasa lebih praktis. Remote Control mempunyai dimensi yang jauh 
lebih kecil dan bersifat portable 
 Dengan keypad, seringkali kita harus memikirkan instalasinya di kotak. PC 
Keyboard seringkali dirasa terlalu besar. Remote Control adalah alternatif untuk 
memecahkan masalah ini. Remote Control yang dibahas saat ini adalah remote control 
merk Sony, di mana remote control ini sudah mempunyai standard bentuk data yang 
mudah diperoleh. 
  

 
Gambar 1 

Teknik Pengkodean Pulsa 
  

 Gambar 1, menunjukkan teknik peng-kode-an data dari remote kontrol Sony, di 
mana dalam hal ini kode 0 diwakili oleh sebuah pulsa positif yang pendek dan kode 1 
diwakili oleh sebuah pulsa positif yang panjang. Pada gambar tersebut, tampak data 5 
(0101) ditransfer secara serial dengan teknik peng-kode-an secara pulsa. 
 Aplikasi ini sebetulnya tidak memerlukan pemancar infrared, karena bagian 
pemancar infrared sudah digantikan oleh remote kontrol Sony. Namun untuk 
menjelaskan bagaimana proses komunikasi data infra merah terjadi, ada baiknya kita 
bahas sedikit mengenai proses pemancaran sinyal infra merah yang dilakukan oleh 
Modul IR-8510. Modul ini selain berfungsi sebagai penerima infra merah, juga 
mempunyai bagian pemancar di mana pancaran data dilakukan dengan bentuk modulasi 
yang mengikuti modulasi remote kontrol infra merah. 
 Modulasi dilakukan dengan frekwensi carrier sebesar 30-40 KHz dengan adanya 
frekwensi carrier sebagai akan membentuk pulsa negatif dan tidak adanya frekwensi 
carrier akan membentuk pulsa positif. (gambar 3) 
  



 
Gambar 2 

Antar Muka IR-8510 
  
 Pada gambar 2 tampak data serial dari TXD dimodulasi oleh U1B yang 
membentuk rangkaian modulator dan diteruskan ke buffer U1C. Data serial akan 
tercampur dengan frekwensi carrier 30-40 KHz dan diteruskan ke penguat arus dari LED 
Infra merah.  
 Pada aplikasi ini, bagian tersebut telah berada dalam remote kontrol infra merah 
yang digunakan. Bagian yang digunakan dalam IR-8510 pada aplikasi ini adalah modul 
IRM8510 (Infrared Receiver Module) yang akan mengubah sinyal-sinyal yang 
termodulasi dari remote kontrol ke pulsa negatif maupun positif dengan band pass filter 
yang ada di dalamnya. 
  

 
Gambar 3 

Timing Diagram 

Gambar 4 

Blok Diagram Sistem 
  
 Pada gambar 4, tampak Modul DST-52 berfungsi sebagai penampil nomor-nomor 
tombol dari remote kontrol melalui sinyal yang dikirimkan oleh remote kontrol SONY ke 
bagian penerima dari Modul IR-8510. Keluaran dari modul ini akan membentuk pulsa-
pulsa seperti pada gambar 1. Modul DST-52 akan menterjemahkan pulsa-pulsa tersebut 
ke dalam kode-kode tombol sesuai table 1. 
  

Tabel 1 
Tabel Fungsi Tombol dan Data yang diterima (sesuai remote Sony tipe RM-827S) 

Tombol Hexa Tombol Hexa 
1 #080 Vol- #093 
2 #081 Power(toggle) #095 
3 #082 PIC Mode #096 
4 #083 A/B #097 
5 #084 TV/Video #0A5 
6 #085 Sleep #0B6 
7 #086 + #0F4 
8 #087 - #0F5 
9 #088 Select #0FC 
0 #089     

Prog+ #090     

Prog- #091     

Vol+ #092     

Gambar 5 

Diagram Alir Utama 
  
 Gambar 5 menunjukkan alur yang terjadi dari aplikasi ini. Program diawali 
dengan proses inisialisasi LCD dan timer. Dalam hal ini fungsi timer digunakan untuk 
menghitung lebar pulsa dari sinyal infra merah. Kemudian dilanjutkan dengan 
pengambilan data remote. Pada proses ini, pulsa-pulsa infra merah dari remote kontrol 
dikodekan menjadi 8 bit data dengan teknik peng-kode-an pulsa. 
 Remote kontrol tipe RM-827S tidak mengenali kode di bawah 80H, oleh karena 
itu agar kode-kode di bawah 80H tidak akan diproses. Kode 80H ke atas akan diproses 
dengan menunjuk ke lokasi-lokasi tertentu dari table nomor dan menampilkannya pada 
layar LCD. 
  
  
Listing 1 
$MOD51 
 CSEG 
;============= 
;HD44780 
;============= 
  
Posisi_Awal   EQU 05CBH              
;Mengatur posisi cursor LCD ke posisi awal 
  
KirimPesan_LCD   EQU 05E0H              
;Mengirim data di alamat yang ditunjuk oleh DPTR ke LCD hingga data 0F 
;- DPTR diisi dengan alamat awal data yang dikirim 
;- Akhir data adalah 0FH 
  
Init_LCD   EQU 05ECH              
;Inisialisasi LCD 



  
Kirim_Karakter   EQU 064AH              
;Mengirim data ke register data LCD 
;- Data diisi di akumulator 
  
  
  
ROM   EQU 2000H 
  
 Org ROM   ;Reset Vector 
 Ajmp Start   ; 
 Org ROM+3H   ;External Interrupt 0 Vector 
 Reti    ; 
 Org ROM+0BH   ;Timer 0 Interrupt Vector 
 Reti 
 Org ROM+13H   ;External Interrupt 1 Vector 
 Reti    ; 
 Org ROM+1BH   ;Timer 1 Interrupt Vector 
 Reti    ; 
 Org ROM+23H   ;Serial Interrupt Vector 
 Reti    ; 
  
  
PanjangData EQU 26 
  
Start: 
 Lcall Init_LCD 
 Acall InitTimer  ;Inisial timer untuk trap data 
Loop: 
 Acall AmbilRemote 
 Cjne A,#7FH,$+3  ;Kode di bawah 80H, abaikan 
 Jc Loop   ; 
 Acall TunjukTabelTombol ;Tunjuk tabel sesuai kode yang diterima 
 Movx A,@DPTR   ;Bila tabel kosong, abaikan 
 Jz Loop   ; 
 Lcall Posisi_Awal  ;Tampilkan isi tabel bila tidak kosong 
 Lcall KirimPesan_LCD  ; 
 Lcall Posisi_Awal  ; 
 Ajmp Loop 
  
  
TunjukTabelTombol: 
 Mov DPTR,#TabelTombol 
 Clr C 
 Subb A,#80H 
 Mov B,#7 
 Mul AB 
  
 Clr C 
 Add A,DPL 
 Jnc TidakIncDPH 
 Inc DPH 
TidakIncDPH: 
 Mov DPL,A 
 Mov A,B 
 Add A,DPH 
 Mov DPH,A 



 Ret   
  
AmbilRemote: 
 Acall CekStartBit  ;Tunggu Start Bit 
 Mov R7,#8   ;Panjang data bit = 8 
  
AmbilData: 
        Push    ACC                       ;Ambil 1 bit 
        Acall   AmbilBit                     ; 
        Pop     ACC                       ; 
 Rrc A   ;Geser ke akumulator 
 Djnz R7,AmbilData  ;Lakukan 8x 
 Clr TR0 
 Ret 
  
CekStartBit: 
 Jb P1.3,$ 
 Setb TR0 
 Jnb P1.3,$ 
 Mov A,TH0 
 Cjne A,#08H,$+3 
 Jnc CekStartBitSelesai 
 Clr TR0 
 Mov TH0,#00 
 Mov TL0,#00 
 Ajmp CekStartBit 
  
CekStartBitSelesai: 
 Clr TR0 
 Mov TH0,#00 
 Mov TL0,#00 
 Ret 
  
AmbilBit: 
 Jb P1.3,$ 
 Setb TR0 
 Jnb P1.3,$ 
 Clr TR0 
 Mov A,TH0 
 Cjne A,#03,$+3 
 Jnc Bit1 
 Clr C 
 Mov TH0,#0 
 Mov TL0,#0 
 Ret 
  
  
Bit1: 
 Setb C 
 Mov TH0,#0 
 Mov TL0,#0 
 Ret 
  
InitTimer: 
 Mov TH0,#00   ; 
 Mov TL0,#00   ; 
        Push    ACC                       ; 



 Mov A,TMOD 
 Anl A,#0F0H 
 Orl A,#01H 
 Mov TMOD,A 
        Pop     ACC 
 Ret 
  
;=========================================== 
; TABEL TOMBOL-TOMBOL REMOTE CONTROL TIPE RM-827S 
; - Data-data yang masih berisi 00 adalah kode cadangan untuk remote 
control 
; - Sony tipe yang lain 
;=========================================== 
  
TabelTombol: 
 DB '1     ',0FH 
 DB '2     ',0FH 
 DB '3     ',0FH 
 DB '4     ',0FH 
 DB '5     ',0FH 
 DB '6     ',0FH 
 DB '7     ',0FH 
 DB '8     ',0FH 
 DB '9     ',0FH 
 DB '0     ',0FH 
 DB 0,0,0,0,0,0,0 ;8a 
 DB 0,0,0,0,0,0,0 ;8b 
 DB '1-    ',0FH  
 DB '2-    ',0FH  
 DB 0,0,0,0,0,0,0 ;8e 
 DB 0,0,0,0,0,0,0 ;8f 
 DB 'Prog+ ',0FH 
 DB 'Prog- ',0FH 
 DB 'Vol+  ',0FH 
 DB 'Vol-  ',0FH 
 DB 0,0,0,0,0,0,0 
 DB 'Power ',0FH 
 DB 'Pic Md',0FH 
 DB 'A/B   ',0FH 
 DB 0,0,0,0,0,0,0 ;98 
 DB 0,0,0,0,0,0,0 ;99 
 DB 0,0,0,0,0,0,0 ;9a 
 DB 0,0,0,0,0,0,0 ;9b 
 DB 0,0,0,0,0,0,0 ;9c 
 DB 0,0,0,0,0,0,0 ;9d 
 DB 0,0,0,0,0,0,0 ;9e 
 DB 0,0,0,0,0,0,0 ;9f 
 DB 0,0,0,0,0,0,0 ;a0 
 DB 0,0,0,0,0,0,0 ;a1 
 DB 0,0,0,0,0,0,0 ;a2 
 DB 0,0,0,0,0,0,0 ;a3 
 DB 0,0,0,0,0,0,0 ;a4 
 DB 'TV/VID',0FH ;a5 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 



 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0  
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
  
 DB 'Sleep ',0FH ;b6 
 DB 0,0,0,0,0,0,0 ;b7 
 DB 0,0,0,0,0,0,0 ;b8 
 DB 0,0,0,0,0,0,0 ;b9 
 DB 0,0,0,0,0,0,0 ;ba 
 DB 0,0,0,0,0,0,0 ;bb 
 DB 0,0,0,0,0,0,0 ;bc 
 DB 0,0,0,0,0,0,0 ;bd 
 DB 0,0,0,0,0,0,0 ;be 
 DB 0,0,0,0,0,0,0 ;bf 
 DB 0,0,0,0,0,0,0 ;c0 
 DB 0,0,0,0,0,0,0 ;c1 
 DB 0,0,0,0,0,0,0 ;c2 
 DB 0,0,0,0,0,0,0 ;c3 
 DB 0,0,0,0,0,0,0 ;c4 
 DB 0,0,0,0,0,0,0 ;c5 
 DB 0,0,0,0,0,0,0 ;c6 
 DB 0,0,0,0,0,0,0 ;c7 
 DB 0,0,0,0,0,0,0 ;c8 
 DB 0,0,0,0,0,0,0 ;c9 
 DB 0,0,0,0,0,0,0 ;ca 
 DB 0,0,0,0,0,0,0 ;cb  
 DB 0,0,0,0,0,0,0 ;cc 
 DB 0,0,0,0,0,0,0 ;cd 
 DB 0,0,0,0,0,0,0 ;ce 
 DB 0,0,0,0,0,0,0 ;cf 
 DB 0,0,0,0,0,0,0 ;d0 
 DB 0,0,0,0,0,0,0 ;d1 
 DB 0,0,0,0,0,0,0 ;d2  
 DB 0,0,0,0,0,0,0 ;d3 
 DB 0,0,0,0,0,0,0 ;d4 
 DB 0,0,0,0,0,0,0 ;d5 
 DB 0,0,0,0,0,0,0 ;d6 
 DB 0,0,0,0,0,0,0 ;d7 
 DB 0,0,0,0,0,0,0 ;d8 
 DB 0,0,0,0,0,0,0  
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 



 DB 0,0,0,0,0,0,0  
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0  
 DB 0,0,0,0,0,0,0 
 DB '+     ',0FH 
 DB '-     ',0FH 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 0,0,0,0,0,0,0 
 DB 'Select',0FH 
 END 
  
 


